if(CMAKE_D_COMPILER_FORCED)
  # The compiler configuration was forced by the user.
  # Assume the user has configured all compiler information.
  set(CMAKE_D_COMPILER_WORKS TRUE)
  return()
endif()

include(CMakeTestCompilerCommon)

# Remove any cached result from an older CMake version.
# We now store this in CMakeCCompiler.cmake.
unset(CMAKE_D_COMPILER_WORKS CACHE)

# This file is used by EnableLanguage in cmGlobalGenerator to
# determine that that selected C compiler can actually compile
# and link the most basic of programs.   If not, a fatal error
# is set and cmake stops processing commands and will not generate
# any makefiles or projects.
if(NOT CMAKE_D_COMPILER_WORKS)
  PrintTestCompilerStatus("D" "")
  file(WRITE ${CMAKE_BINARY_DIR}${CMAKE_FILES_DIRECTORY}/CMakeTmp/testDCompiler.d
    "void main() {}")
  try_compile(CMAKE_D_COMPILER_WORKS ${CMAKE_BINARY_DIR}
    ${CMAKE_BINARY_DIR}${CMAKE_FILES_DIRECTORY}/CMakeTmp/testDCompiler.d
    OUTPUT_VARIABLE __CMAKE_D_COMPILER_OUTPUT)
  # Move result from cache to normal variable.
  set(CMAKE_D_COMPILER_WORKS ${CMAKE_D_COMPILER_WORKS})
  unset(CMAKE_D_COMPILER_WORKS CACHE)
  set(D_TEST_WAS_RUN 1)
endif()

if(NOT CMAKE_D_COMPILER_WORKS)
  PrintTestCompilerStatus("D" " -- broken")
  file(APPEND ${CMAKE_BINARY_DIR}${CMAKE_FILES_DIRECTORY}/CMakeError.log
    "Determining if the D compiler works failed with "
    "the following output:\n${__CMAKE_D_COMPILER_OUTPUT}\n\n")
  message(FATAL_ERROR "The D compiler \"${CMAKE_D_COMPILER}\" "
    "is not able to compile a simple test program.\nIt fails "
    "with the following output:\n ${__CMAKE_D_COMPILER_OUTPUT}\n\n"
    "CMake will not be able to correctly generate this project.")
else()
  if(D_TEST_WAS_RUN)
    PrintTestCompilerStatus("D" " -- works")
    file(APPEND ${CMAKE_BINARY_DIR}${CMAKE_FILES_DIRECTORY}/CMakeOutput.log
      "Determining if the D compiler works passed with "
      "the following output:\n${__CMAKE_D_COMPILER_OUTPUT}\n\n")
  endif()

  # Try to identify the ABI and configure it into CMakeCCompiler.cmake
  #include(${CMAKE_ROOT}/Modules/CMakeDetermineCompilerABI.cmake)
  #CMAKE_DETERMINE_COMPILER_ABI(D ${CMAKE_ROOT}/Modules/CMakeCCompilerABI.c)
  # Try to identify the compiler features
  include(${CMAKE_ROOT}/Modules/CMakeDetermineCompileFeatures.cmake)
  CMAKE_DETERMINE_COMPILE_FEATURES(D)

  # Re-configure to save learned information.
  # TODO: fix path for upstream
  configure_file(
    cmake/Modules/CMakeDCompiler.cmake.in
    ${CMAKE_PLATFORM_INFO_DIR}/CMakeDCompiler.cmake
    @ONLY
    )
  include(${CMAKE_PLATFORM_INFO_DIR}/CMakeDCompiler.cmake)
endif()

unset(__CMAKE_D_COMPILER_OUTPUT)
